

Počítačový algebraický systém Maple jako pomůcka při studiu
předmětu Matematika I a II.

Primitivní funkce, určitý integrál, nevlastní integrály

Program Maple může být velmi dobrým pomocníkem při hledání primitivních funkcí i při výpočtu určitých integrálů. Přesto se neobejdeme bez dobré znalosti teorie výpočtů primitivních funkcí a určitých integrálů. Bez těchto znalostí může dojít ke špatné interpretaci výsledků, nebo se nám nepodaří získat výsledek, i když po určitých úpravách lze výsledek získat.

Příkaz pro výpočet primitivní funkce v Maple má tvar:

```
int(f(x),x);
```

Příkaz pro výpočet určitého integrálu $\int_a^b f(x)dx$ v Maple má tvar:

```
int(f(x),x=a..b);
```

- **Příklad 1:** Vypočtěte primitivní funkci k funkci $f(x) = x^5$

```
int(x^5,x);
```

Maple nám vrátí výsledek

$$\frac{x^6}{6}$$

- **Příklad 2:** Vypočtěte určitý integrál $\int_0^1 x^5 dx$

```
int(x^5,x=0..1);
```

Maple nám vrátí výsledek

$$\frac{1}{6}$$

Teď vyzkoušíme složitější příklad.

- **Příklad 3:** Vypočtěte primitivní funkci k funkci $f(x) = \frac{1}{\sqrt{x}(2+2\sqrt{x}+x)}$

```
int(1/(sqrt(x)*(2+2*sqrt(x)+x)),x);
```

Maple nám vrátí výsledek

$$2 \arctan(\sqrt{x} + 1)$$

O správnosti výsledku se můžeme přesvědčit derivací:

```
diff(%,x);
```

Maple nám vrátí výsledek

$$\frac{1}{\sqrt{x}(2+2\sqrt{x}+x)}$$

Chceme-li zapsat výsledek i se zadáním, můžeme využít příkaz `Int` následujícím způsobem:

```
Int(1/(sqrt(x)*(2+2*sqrt(x)+x)),x)=int(1/(sqrt(x)*(2+2*sqrt(x)+x)),x);
```

Maple nám vrátí výsledek

$$\int \frac{1}{\sqrt{x}(2+2\sqrt{x}+x)} dx = 2 \arctan(\sqrt{x}+1)$$

Dále si ukážeme integrály, na které si musíme dát pozor.

- **Příklad 4:** Vypočítejte primitivní funkci a k funkci $f(x) = x^n$ a určitý integrál $\int_0^1 x^n dx$

`int(x^n,x);`

Maple nám vrátí výsledek

$$\frac{x^{(n+1)}}{n+1}$$

`int(x^n,x=0..1);`

Maple nám vrátí výsledek

$$\lim_{x \rightarrow 0+} -\frac{x^{(n+1)} - 1}{n+1}$$

Musíte si uvědomit, že výsledek platí pouze pro $n \neq -1$. Poslední výsledek můžeme zlepšit, jestliže použijeme podmínku pro n .

`assume(n>-1);`

`int(x^n,x=0..1);`

Maple nám tentokrát vrátí výsledek

$$\frac{1}{n+1}$$

nebo

`assume(n<-1);`

`int(x^n,x=0..1);`

Maple nám tentokrát vrátí výsledek

$$\infty$$

Ale ani výpočet primitivní funkce k funkci $f(x) = \frac{1}{x}$ není bez problémů. Musíme si uvědomit, že Maple počítá v oboru komplexních čísel.

`int(1/x,x);`

Maple nám vrátí (počítáme-li v oboru reálných čísel) špatný výsledek

$$\ln(x)$$

Správný výsledek dostaneme až příkazem **Re** (reálná část):

`Re(int(1/x,x;))`

$$\ln(|x|)$$

Ted' si ukážeme, jak si poradit, když nám Maple primitivní funkci nevypočte.

- **Příklad 5:** Vypočítejte primitivní funkci k funkci $f(x) = \frac{x \arcsin(x)}{\sqrt{1-x^2}}$

`int(x*arcsin(x)/sqrt((1-x^2)),x);`

Maple nám nevrátí uspokojivý výsledek

$$\int \frac{x \arcsin(x)}{\sqrt{1-x^2}} dx$$

Zde můžeme použít příkaz pro výpočet primitivní funkce pomocí metody per partes. Příkaz má tvar:

`student[intparts](Int(f(x),x),u(x));`,

kde $f(x)$ je integrovaná funkce a $u(x)$ je funkce, kterou chceme v metodě per partes derivovat. V našem případě tedy použijeme příkaz:

`student[intparts](Int(x*arcsin(x)/sqrt((1-x^2),x),arcsin(x));`

Jako funkci, kterou budu derivovat jsme zvolili $\arcsin(x)$. Maple nám vrátí výsledek

$$-\arcsin(x)\sqrt{1-x^2} - \int -1 dx$$

Zde je integrál ve výsledku jednoduchý a lehce jej vypočteme. Pokud je integrál složitější, musíme pokračovat ve výpočtech dílčích integrálů.

Nyní si ukážeme další funkci, pro kterou nám Maple nedá primitivní funkci, i když ji lze lehce spočítat.

- **Příklad 6:** Vypočítejte primitivní funkci k funkci $f(x) = \frac{2x + \frac{1}{2\sqrt{x}}}{(x^2 + \sqrt{x})^2 + 1}$;

`int((2*x+1/(2*sqrt(x)))/((x^2+sqrt(x))^2+1),x);`

Maple tuto primitivní funkci nespočte. My však vidíme, že můžeme použít substituci $y = \sqrt{x}$. Použijeme tedy substituci a zkusíme primitivní funkci znova spočítat pomocí Maple. Pro substituci můžeme použít příkaz `student[changevar](s,Int(f(x),x),y)`, kde s je substituce $y = \varphi(x)$, $f(x)$ je původní integrovaná funkce a y je nová proměnná.

V našem případě tedy napíšeme:

`F:=student[changevar](y=sqrt(x),Int((2*x+1/(2*sqrt(x)))/((x^2+sqrt(x))^2+1),x),y);`

Dostaneme integrál po substituci:

$$F := \int \frac{4y^3 + 1}{y^8 + 2y^5 + y^2 + 1} dy$$

Nyní musíme ještě integrál vypočítat:

`value(F);`

Dostaneme následující výsledek:

$$\arctan(y^4 + y)$$

Nakonec vrátíme substituci:

`subs(y=sqrt(x),%);`

Maple nám dá již správný výsledek

$$\arctan(x^2 + \sqrt{x})$$

O správnosti se přesvědčíme derivací:

`diff(%,x);`

$$\frac{2x + \frac{1}{2\sqrt{x}}}{(x^2 + \sqrt{x})^2 + 1}$$

Pozorný čtenář si zde všimne, že jsme mohli použít lepší substituci a to $y = x^2 + \sqrt{x}$ a výpočet by byl ještě jednodušší. Záměrně jsme zvolili substituci $y = \sqrt{x}$, protože ta by měla napadnout každého studenta, který prošel základním kurzem výpočtu primitivních funkcí pomocí substitute.

Nyní si ukážeme, jak získáme přibližnou hodnotu určitého integrálu v případě, že Maple integrál nevypočítá analyticky. Použijeme příkaz, který využívá k výpočtu určitého integrálu přibližné numerické metody .

• **Příklad 7:** Vypočtete určitý integrál $\int_0^1 e^{\frac{x^2}{x^2+1}} dx$

```
int(exp(x^2/(x^2+1)),x=0..1);
```

Maple nám vrátí pouze opis integrálu:

$$\int_0^1 e^{\frac{x^2}{x^2+1}} dx$$

Použijeme tedy příkaz pro výpočet určitého integrálu, který využívá přibližných numerických metod. Bohužel, jaké metody přesně používá se nedozvíme. Příkaz má tvar `evalf(Int(f(x),x=a..b))`. Implicitně Maple počítá s přesností $\varepsilon = 0.5 \cdot 10^{1-digits}$, kde *digits* je hodnota proměnné *Digits*, což je počet míst, na které Maple počítá. Uživatel si tuto proměnnou může nastavit. Použijeme tedy přibližný numerický výpočet:

```
evalf(Int(exp(x^2/(x^2+1)),x=0..1));
```

Maple nám vrátí výsledek:

1.255621168

Maple si poměrně dobře poradí i s nevlastními integrály.

• **Příklad 8:** Vypočtete určitý integrál $\int_1^{\infty} \frac{1}{t^2+2+\sin(t)} dt$

Tento integrál konverguje, protože $0 \leq \frac{1}{t^2+2+\sin(t)} \leq \frac{1}{t^2}$ a $\int_1^{\infty} \frac{1}{t^2} dt = 1$, tedy konverguje. Protože primitivní funkci k integrované funkci neumíme ani my ani Maple vypočítat, použijeme hned přibližný numerický výpočet.

```
evalf(Int(1/(t^2+2+sin(t)),t=1..infinity));
```

Maple nám vrátí následující výsledek:

.6241667108

Pozor nesmíte použít příkaz:

```
evalf(Int(1/(t^2+2+sin(t)),t=1..100000000));
```

Maple nám vrátí následující výsledek:

40296.74067

což je samozřejmě špatný výsledek. Použijeme-li příkaz:

```
evalf(Int(1/(t^2+2+sin(t)),t=1..100000100));
```

dostaneme výsledek:

40297.14364

Rozdíl posledních dvou výsledků je 0.40297 ale následující příkaz nám dá jiný výsledek:

```
evalf(Int(1/(t^2+2+sin(t)),t=100000000..100000100));
```

dostaneme výsledek:

-.8999900001e-7

Vše je způsobeno tím, že pracujeme s přesností na deset desetinných míst a předchozí výsledky jsou zatíženy chybou počítače. Nyní použijeme stejný příkaz pro určitý integrál, který nekonverguje.

• **Příklad 9:** Vypočtěte určitý integrál $\int_1^{\infty} \frac{1}{t+2+\sin(t)} dt$

Tento integrál diverguje, protože $\frac{1}{t+2+\sin(t)} \geq \frac{1}{t+3}$ (pro $t > 1$) a $\int_1^{\infty} \frac{1}{t+3} dx = \infty$, tedy diverguje. Protože primitivní funkci k integrované funkci neumíme opět určit, použijeme přibližný numerický výpočet.

```
evalf(Int(1/(t+2+sin(t)),t=1..infinity));
```

Maple nám vrátí následující výsledek:

Float(∞)